

# MAPLE Notes for MACM 204

Michael Monagan  
Department of Mathematics  
Simon Fraser University  
August, 2012.

```
> restart;
```

These notes are for Maple 13. They are platform independent, i.e., they are the same for the Macintosh, PC, and Unix versions of Maple. These notes should be backwards compatible with Maple versions 10, 11, 12, and forwards compatible with Maple 14, 15, 16.

## Maple as a Graphing Calculator

Input of a numerical calculation uses +, -, \*, /, and ^ for addition, subtraction, multiplication, division, and exponentiation respectively.

```
> 1+2;
```

3

```
> 2*6;
```

12

```
> 2^3;
```

8

```
> 4-2*3;
```

-2

Observe that every command ends with a semicolon ; This is a grammatical requirement of Maple. If you forget, Maple will assume that the command is not complete. This allows you to break long commands across a line. For example

```
> 1+2*3/  
  (2+3);
```

$\frac{11}{5}$

Notice that the output is an exact rational number and not the decimal number 2.2. Here is another example

```
> 120/105;
```

$\frac{8}{7}$

Because the input involved integers, not decimal numbers, Maple calculates the exact fraction when there is a division, automatically cancelling out the greatest common divisor (GCD). In this case the GCD is 15, which you can calculate specifically as

```
> igcd(120,105);
```

15

Here is how you would do some decimal calculations. The presence of a decimal point . in a number means that the number is a decimal number and Maple will, by default, do all calculations to 10 decimal places.

```
> 120/105.0;
```

1.142857143

```
> 4./3.;
```

1.333333333

```
> sqrt(2), sqrt(4), sqrt(8);
```

$\sqrt{2}$ , 2,  $2\sqrt{2}$

```
> sqrt(2.0), sqrt(4.0), sqrt(8.0);
```

1.414213562, 2.000000000, 2.828427125

```
> exp(0), exp(1), exp(2);
```

1, e, e<sup>2</sup>

```
> exp(0.0), exp(1.0), exp(2.0);
```

1., 2.718281828, 7.389056099

Notice the difference caused by the presence of a decimal point in these examples.

Now, if you have input an exact quantity, like the  $\sqrt{2}$  above, and you now want to get a numerical value, use the evalf command to evaluate to floating point. Use the % character to refer to the previous Maple output.

```
> sqrt(2);
```

$\sqrt{2}$

```
> evalf(%);
```

1.414213562

By default you get 10 decimal digits. Maple is like an HP calculation using 10 digit arithmetic. If you want a value to higher precision, you can set the value of the Maple variable Digits first.

```
> Digits := 50;
```



We are going to use this polynomial for a few calculations. We want to give it the name  $f$  so we can refer to it later. We do this using the assignment operation in Maple as follows. If you like, think of  $f$  as a programming variable. But  $x$  is still an unknown.

```
> f := x^4-3*x+2;
```

$$f := x^4 - 3x + 2$$

The name  $f$  is now a variable. It refers to the polynomial. Here is its value and its derivative.

```
> f;
```

$$x^4 - 3x + 2$$

```
> diff(f,x);
```

$$4x^3 - 3$$

To evaluate  $f$  this as a function at the point  $x = 3$  use the eval command as follows

```
> eval(f,x=3);
```

$$74$$

The following commands factor  $f$  into irreducible factors over the field of rational numbers and then compute 10 digit numerical approximations to the real roots respectively.

```
> factor(f);
```


$$(x-1)(x^3+x^2+x-2)$$

```
> fsolve(f=0,x);
```

$$0.8105357138, 1.$$

You can graph functions using the plotting commands. The basic syntax for the **plot** command for a function of one variable is illustrated as follows:

```
> plot(f,x=0.2 .. 1.3);
```



In the graph I can see a local minimum near  $x=0.9$ . We can find this point using calculus. The command `fsolve( f(x)=0, x )`, on input of a polynomial  $f(x)$  computes 10 digit numerical approximations for the real roots of  $f(x)$ . `solve` gives you an exact formula for all the roots.

```
> fsolve(diff(f,x)=0,x);
```

```
0.9085602964
```

```
> solve(diff(f,x)=0,x);
```

$$\frac{6^{1/3}}{2}, -\frac{6^{1/3}}{4} + \frac{1}{4} I\sqrt{3} 6^{1/3}, -\frac{6^{1/3}}{4} - \frac{1}{4} I\sqrt{3} 6^{1/3}$$


Here are the decimal approximations for these formulae. So first one is the one that `fsolve` computed is the only real root.

```
> evalf(%);
```

```
0.9085602965, -0.4542801482 + 0.7868362978I, -0.4542801482
-0.7868362978I
```

Another way to do this is to graph the function and its derivative on the same graph. I've used the `thickness = 3` option to draw thicker lines so we can see the curves more clearly. Also objects of the form `[f1,f2,f3]` are called lists in Maples.


```
> plot( [f,diff(f,x)], x=0.5..1.3, thickness=3 );
```




Exercise: Try to graph the rational function  $\frac{1}{x^3 - 2x + 1}$  and calculate any local minima or maxima.

Some other standard functions

```
> plot( [ln(x),sqrt(x),exp(x)], x=-3..3, y=-4..4, thickness=2,  
numpoints=100 );
```



```
> plot( [sin(x),cos(x),tan(x)], x=-10..10, y=-10..10, thickness=2  
);
```



We have used the name `f` as variable to refer to formulae and the symbols `x` for an unknown in a formula. Often you will have assigned to a name like we have done here to `f` but you want now to use the name `f` as a symbol again, not as a variable. You can unassign the value of a name as follows

```
> f;
```

$$x^4 - 3x + 2$$

```
> f := 'f';
```

$$f := f$$

```
> f;
```

$$f$$

## Derivatives and integrals

```
> f := x^2;
```

$$f := x^2$$

Here is the derivative and antiderivative of `f(x)` wrt `x`.

```
> diff(f,x);
```

$$2x$$

```
> int(f,x);
```

$$\frac{x^3}{3}$$

Here are another couple of standard examples

```
> g := 1/sqrt(4-x^2);  
h := x/(1-x^2);
```

$$g := \frac{1}{\sqrt{4-x^2}}$$

$$h := \frac{x}{1-x^2}$$

```
> int(g,x);
```

$$\arcsin\left(\frac{x}{2}\right)$$

```
> int(h,x);
```

$$-\frac{1}{2} \ln(x-1) - \frac{1}{2} \ln(x+1)$$

Notice that Maple does not include a constant C of integration. It seems all the computer algebra systems have adopted this convention for simplicity. To

compute a definite integral  $\int_a^b f(x) dx$  the Maple command is **int(f(x),x=a..b)**. For

example

```
> int(f,x=0..1);
```

$$\frac{1}{3}$$

```
> int(g,x=0..1);
```

$$\frac{\pi}{6}$$

Maple can differentiate any formula but it cannot find closed form formulas for every function. Here are some examples

```
> f := x*sin(x);
```

$$f := x \sin(x)$$

```
> int(f,x);
```

$$\sin(x) - x \cos(x)$$

```
> f := sin(x)/x;
```

$$f := \frac{\sin(x)}{x}$$

```
> int(f,x);
```

$$\text{Si}(x)$$

Huh, what's that? It's one of the many special functions that Maple "knows" called

the sine integral. Let's check it

```
> diff(%,x);
```

$$\frac{\sin(x)}{x}$$

Here's one that Maple cannot do - took me a while to find one.

```
> f := sin(x^2)*ln(1+x);
```

$$f := \sin(x^2) \ln(1+x)$$

```
> int(f,x);
```

$$\int \sin(x^2) \ln(1+x) dx$$

```
> diff(%,x);
```


$$\sin(x^2) \ln(1+x)$$

```
> cons := int(f,x=1/2..3/2);
```

$$cons := \int_{\frac{1}{2}}^{\frac{3}{2}} \sin(x^2) \ln(1+x) dx$$

Now this value is a constant. If we graph the function  $f$  on  $[1/2, 3/2]$  we can see that it is smaller than 1. To get a numerical approximation use `evalf`.

```
> plot(f,x=0..2);
```



```
> evalf(cons);
```

0.5350415599

## Loops, sequences, lists and sets

Here is a simple example of a for loop that computes the sum of the first 5 integers

```
> s := 0;
  for k from 1 to 5 do
    s := s+k;
  od;

s:= 0
s:= 1
s:= 3
s:= 6
s:= 10
s:= 15

> s;

15
```

Here is another simple loop to print out the prime between 100 and 110 that counts through the odd numbers

```
> for i from 101 to 110 by 2 do
  if isprime(i) then printf("%d is prime\n",i) end if;
od;
101 is prime
103 is prime
107 is prime
109 is prime
```

The Maple command for representing a definite integral without computing it is `Int(f(x),x=a..b)`. Compare

```
> Int( x^2, x=0..1 );


$$\int_0^1 x^2 dx$$


> int( x^2, x=0..1 );


$$\frac{1}{3}$$

```

Here is a loop to compute some integrals

```
> for i from 1 to 4 do
  Int(x^i,x=0..1) = int(x^i,x=0..1);
```

```
od;
```

$$\int_0^1 x \, dx = \frac{1}{2}$$

$$\int_0^1 x^2 \, dx = \frac{1}{3}$$

$$\int_0^1 x^3 \, dx = \frac{1}{4}$$

$$\int_0^1 x^4 \, dx = \frac{1}{5}$$

Also useful is the **sum( f(i), i=a..b )** command for computing formulas for sums. Here is the sum of the first n positive integers 1+2+...+n

```
> Sum( k, k=1..n );
```

$$\sum_{k=1}^n k$$

```
> sum( k, k=1..n );
```

$$\frac{(n+1)^2}{2} - \frac{n}{2} - \frac{1}{2}$$

```
> factor(%);
```

$$\frac{n(n+1)}{2}$$

```
> for i from 1 to 4 do  
  Sum(k^i, k=1..n) = factor(sum(k^i, k=1..n));  
od;
```

$$\sum_{k=1}^n k = \frac{n(n+1)}{2}$$

$$\sum_{k=1}^n k^2 = \frac{n(n+1)(2n+1)}{6}$$

$$\sum_{k=1}^n k^3 = \frac{n^2(n+1)^2}{4}$$

$$\sum_{k=1}^n k^4 = \frac{n(n+1)(2n+1)(3n^2+3n-1)}{30}$$

Exercise: Try to write a loop that compute  $1 + x + \frac{x^2}{2} + \frac{x^3}{3!} + \frac{x^4}{4!} + \frac{x^5}{5!}$ .

A sequence in Maple is just a "sequence of values separated by commas". For example

```
> 2,1,3,2;
2, 1, 3, 2
```

To create an unordered list of values, put a sequence inside [ ] brackets. To create a set of values, with duplicates removed, put a sequence in { } brackets.

```
> L := [2,1,3,2];
L:= [2, 1, 3, 2]
```

```
> S := {2,1,3,2};
S:= {1, 2, 3}
```

You can create sequences, lists, and sets of any values, of formulas, matrices, not just numbers.

```
> L := [sin(x), cos(x), tan(x)];
L:= [sin(x), cos(x), tan(x)]
```

You can access the i'th element of a list or set using the subscript notation L[i] or S[i]. The number of elements in a list or set is given by nops(S) or nops(L).

```
> L[1];
sin(x)
```

```
> nops(L);
3
```

```
> L[4];
Error, invalid subscript selector
```

Here are the derivatives of the elements of the list

```
> for i to nops(L) do
  diff(L[i],x);
od;
cos(x)
-sin(x)
1 + tan(x)2
```

For more operations on lists and sets see ?list and ?set

The command **seq( f(i), i=a..b )** creates a sequence

```
> seq( binomial(6,i), i=0..6 );
1, 6, 15, 20, 15, 6, 1
```

```
> seq( xi, i=1..4 );
x, x2, x3, x4
```

Using seq we can create a sequence of functions which we could plot together

```
> F := [seq( 1-xi, i=1..4 )];
F:= [1 - x, 1 - x2, 1 - x3, 1 - x4]
```

```
> P := [seq( ithprime(i), i=1..10 )];
P:= [2, 3, 5, 7, 11, 13, 17, 19, 23, 29]
```

└ Exercise: Can you create the sequence  $x, -x^3, x^5, -x^7, x^9$ .